

Calling HP macros from WP

In this zipped file are:

HPMAC.DOC	This document.
DEFHPMAC.WPM	The macro which defines the HP macro.
CALHPMAC.WPM	The macro which sets the calling code in the document to be merged.
VALIDAT.WPM	A numerical validating macro for exteriorising this code if you wish to restructure the macros (see below).

General Principles

Start with the full document including any graphics, tables etc. Anything that will print to a LaserJet can be saved as an HP macro. Now move the cursor to the start of the text and/or graphic you want to download as an HP macro and execute DEFHPMAC. This will insert the start HP macro definition at this point in the document. It inserts the stop definition in a footer the code of which is placed at the top of the current page as it is the only way to be certain that the stop definition comes out last. WP has a nasty habit of re-arranging the order in which it sends data to the printer -ie it tends to lump all the graphics together irrespective of their order in the document. Now print this document to the printer - it will produce an empty page but don't panic. In fact if it doesn't throw a page there is probably something wrong. You will probably want to save this document for future use as the HP macro disappears on powerdown. You only need to create this document once for each macro and just recall and print it after powering up the printer. If you are using a local printer you could include document retrieve and print as part of startup macro. On a network you could include it as part of the startup procedure.

To recall the HP macro, just execute CALHPMAC in the document which requires inclusion of the HP macro. It will insert the calling code in the document as a printer command. This is not position dependent. The code needs to be placed on every page which requires the information to be printed. If it is needed on every page you might try putting it in a header. Remember also that HP macros are like overlays so your document needs to account for the space occupied by the text or graphics in the HP macro. This is OK for a macro placed at the top of the page as you can use Advance to set it to the correct position. If you want to get fancy and put a graphic in the middle of the page you need to create an empty figure in the same position in the document which calls the macro.

When you run these macros you will be prompted for a Macro ID number as it is possible to store several HP macros. Any number from 0 to 32767 is valid. Obviously it has to be the same for both the defining and calling macros. In most cases you should be able to select any number in this range but be aware that some cartridges contain HP macros and you should avoid this possible conflict. My guess is that it unlikely that you have such a cartridge installed.

Finally it is quite a good idea to work with reveal codes on while you are creating the defining document(s).

Some possible problem areas.

If you use DEFHPMAC in the simplest possible way just to create a letterhead macro then you may not have too many problems. It will just take all the information between the top of page to the bottom and store it as an HP macro. However you should be aware that not all of the information in the form of codes is actually sent to the printer. WordPerfect attempts to reduce the information sent to the printer by only sending changes. This means that even if you have inserted a font code inside the HP definition it will not be sent to the printer if the it is the same as the current font. So it is no good inserting Univers 10 pt if the current font is already Univers 10 pt. The trick here is to define a **different** point size outside the definition to force WP to insert an explicit font selection.

The other essential information that needs to be in an HP macro is vertical and horizontal positioning. You will only have this problem if you are start the macro definition other than at the top of the document. Below are a few problems that can occur and hints for getting it right.

Suppose you have a simplified document:

Line 1
Line 2
Line 3

and you want only lines 2 and 3 to be stored as a macro.

If you start DEFHPMAC at the beginning of Line 2 you will only have line 3 stored in HP macro. **Start it at the end of Line 1.**

If you have two or more HRts separating Lines 1 and 2 ie:

Line 1

Line 2
Line 3

horizontal positioning is not inserted at the beginning of the HP macro although vertical positioning is. This means that when the HP macro is recalled in another document 'Line' in Line 2 will start at the last horizontal positioning in the document. This can look very odd eg

Line2

Line 3

To avoid this it is necessary to force WP to put out a horizontal position. The simplest way to achieve this is to insert a character (any will do) at the beginning of line above Line 2 and start DEFHPMAC to the right of it. In Reveal Codes it will look like:

```
X[Ptr Cmnd][HRt]  
Line 2[HRt]  
Line 3[HRt]
```

where X is the inserted character (I usually use a period).

October 23, 1991 HPMAC.DOC

This may seem like black magic but it all becomes clear if you print the file to disk and disassemble the PCL. There is a useful utility DUMPPCL on the HP forum which makes this much easier if you have to go to that extreme.

DON'T PANIC you may never see any of these problems except, maybe, for the font problem.

Getting Clever

DEFHPMAC allows you to store Header and Footer information as an HP macro. To do this, create the header as you would normally, move to the left (now you see why I said to use Reveal Codes) and run DEFHPMAC. It will have set up a Footer A at the top of the document with the end HP macro definition. If you want to include a footer you need to edit Footer A and include your text etc **before** the printer command. However you should note that when stored as an HP macro the text will save the current position information. If the document which calls the macro has different page margins, the text retrieved from the HP macro will be misaligned. Also if you have a Page numbering style in the footer it will not work. This is only useful for static and fixed information.

To recall the macro in a document you can either run CALHPMAC at the top of each page where it is required or if you want it to appear on every page you can define a header or footer in the calling document and run CALHPMAC while you are in the header/footer edit screen. Including the macro call command in a footer is slightly more preferable. Remember that this is only a way of forcing the macro call on each page. However, you can include other information such as page numbering, a date code etc.

Modifying the macros

If you are setting these up as service macros you may wish to avoid the macro ID prompt altogether and just assign a number automatically. To achieve this you will need to edit the macro and change the statement:

{ASSIGN}MID~~ to {ASSIGN}MID~n~ where n is a number between 0 and 32767.

This must be done in both macros. Listings of the macros are appended to this document for reference.

You will probably wish to rename CALHPMAC to something more meaningful like LETTER. You may wish to define more than one HP macro, say for different letter heads. You could create a document with all the definitions provided they were separated by a Hard Page with each definition on having a different ID.

You could then have several calling macros like LETTER1, LETTER2 etc. If you are likely to have several calling macros you might like to try some drastic surgery on the macros to save some space. If you refer to the listings you will see that the VALIDAT code is repeated. VALIDAT just checks to see that the ID is a valid number. I have included this code in a separate macro VALIDAT.WPM. If you wished you could change the {CALL}validat~ statements to {NEST}validat~ and delete the VALIDAT code from the defining and calling macros. This is probably only worth doing if you have several calling macros.

Let me know if you have any problems or suggestions.

John Filshie (100016,1311)
AFRC Computing Division
West Common,
Harpenden

October 23, 1991 HPMAC.DOC

Hertfordshire

AL5 2JE

England

October 23, 1991 DEFHPMAC

```
{DESCRIPTION}Define an HP macro~
{;}      DEFHPMAC      Ver 1.1  J Filshie  21 October 1991
        Macro to define an HPIII macro.  It should be executed immediately
        in front of the graphic, table or text to be included in the macro.
        Everything from this point to the end of the WP page will be
        included in the WP macro.~

{ASSIGN}MID~~      {;}assign this to any number between 0 - 32767
                   make sure that this is the same as the calling macro
                   a null assignment will request the ID from the user~

{IF EXISTS}MID~
  {GO}chkid~
{END IF}

{LABEL}start~
{TEXT}MID~Enter ID number for HP macro to be defined: ~

{LABEL}chkid~
{ASSIGN}0~{VARIABLE}MID~~
{CALL}validat~
{IF}!{VARIABLE}numeric~~
  {BELL}{PROMPT}ID must be a number - Try again!~{WAIT}20~
  {GO}start~
{END IF}

{DISPLAY OFF}{Format}oppc<27>&f{VARIABLE}MID~Y<27>&f0X{Enter}
{Enter}{Enter}{Enter}
{Goto}{SYSTEM}page~{Enter}
{Format}pfap{Format}oppc<27>&f1X<27>&f10X{Enter}
{Enter}{Enter}{Enter}{Exit}{Enter}{Enter}
{RETURN}

{;}      <<< validat >>>~
{LABEL}validat~
{IF EXISTS}0~
  {ASSIGN}len~{LEN}0~~
  {FOR}i~0~{VARIABLE}len~-1~1~
    {ASSIGN}ch~{MID}0~{VARIABLE}i~{VARIABLE}len~~~
    {IF}{VARIABLE}ch~>47|{VARIABLE}ch~<58~
      {ASSIGN}numeric~-1~
    {ELSE}
      {ASSIGN}numeric~0~
    {END IF}
  {END FOR}
{ELSE}
  {ASSIGN}numeric~0~
{END IF}
{RETURN}      {;}return from validat~
```

October 23, 1991 CALHPMAC

```
{DESCRIPTION}Call an HP macro~  
{;} CALHPMAC Ver 1.1 J Filshie 21 October 1991  
Macro to call an HP macro previously defined by use of DEFHPMAC.  
I think the position of the command is immaterial but am not sure~
```

```
{ASSIGN}MID~~ {;}assign this to any number between 0 - 32767  
make sure that this is the same as the defining macro  
a null assignment will request the ID from the user~
```

```
{IF EXISTS}MID~  
{GO}chkid~  
{END IF}
```

```
{LABEL}start~  
{TEXT}MID~Enter the ID of the HP macro to be called: ~
```

```
{LABEL}chkid~  
{ASSIGN}0~{VARIABLE}MID~~  
{CALL}validat~  
{IF}!{VARIABLE}numeric~~  
{BELL}{PROMPT}ID must be a number - Try again!~{WAIT}20~  
{GO}start~  
{END IF}
```

```
{DISPLAY OFF}{Format}oppc<27>&f{VARIABLE}MID~Y<27>&f3X{Enter}  
{Enter}{Enter}{Enter}  
{RETURN}
```

```
{;} <<< validat >>>~  
{LABEL}validat~  
{IF EXISTS}0~  
{ASSIGN}len~{LEN}0~~  
{FOR}i~0~{VARIABLE}len~-1~1~  
{ASSIGN}ch~{MID}0~{VARIABLE}i~{VARIABLE}len~~~  
{IF}{VARIABLE}ch~>47|{VARIABLE}ch~<58~  
{ASSIGN}numeric~-1~  
{ELSE}  
{ASSIGN}numeric~0~  
{END IF}  
{END FOR}  
{ELSE}  
{ASSIGN}numeric~0~  
{END IF}  
  
{RETURN} {;}return from validat~
```